# SPICE for Circuits and Electronics Using PSpice®

## Second Edition

Muhammad H. Rashid

*Ph. D., Fellow IEE*
*Professor of Electrical Engineering*
*Purdue University at Fort Wayne*

*Prentice Hall, Englewood Cliffs, New Jersey 07632*

chapter

# 8

# Bipolar Junction Transistors

## 8-1 INTRODUCTION

A bipolar junction transistor (BJT) may be specified by a device statement in conjunction with a model statement. Similar to diode models, the BJT model incorporates an extensive range of characteristics: for example, dc and small-signal behavior, temperature dependency, and noise generation. The model parameters take into account temperature effects, various capacitances, and the physical properties of semiconductors.

## 8-2 BJT MODEL

PSpice generates a complex model for BJTs. The model equations that are used by PSpice are described in Gummel and Poon [1] and Getreu [2]. If a complex model is not necessary, the model parameters can be ignored by the users, and PSpice assigns default values to the parameters.

The PSpice model, which is based on the integral charge-control model of Gummel and Poon [1, 6], is shown in Fig. 8-1. The small-signal and static models that are generated by PSpice are shown in Figs. 8-2 and 8-3, respectively.

The model statement for NPN transistors has the general vorm

```
.MODEL QNAME NPN (P1=A1 P2=A2 P3=A3 ...PN=AN)
```

and the general form for PNP transistors is

```
.MODEL QNAME PNP (P1=A1 P2=A2 P3=A3 ...PN=AN)
```




**Figure 8-1** PSpice BJT model.




**Figure 8-2** Small-signal BJT model.




**Figure 8-3** Static BJT model.

where QNAME is the name of the BJT model. NPN and PNP are the type symbols for NPN and PNP transistors, respectively. QNAME, which is the model name, can begin with any character, and its word size is normally limited to eight characters. P1, P2, . . . and A1, A2, . . . are the parameters and their values, respectively. Table 8-1 shows the model parameters of BJTs. If certain parameters are not specified, PSpice assumes the simple Ebers-Moll model [3], which is shown in Fig. 8-4(a).

**TABLE 8-1** MODEL PARAMETERS OF BJTS

| Name | Area | Model parameters | Units | Default | Typical |
|---|---|---|---|---|---|
| IS | * | *p-n* saturation current | Amps | 1E−16 | 1E−16 |
| BF | | Ideal maximum forward beta | | 100 | 100 |
| NF | | Forward current emission coefficient | | 1 | 1 |
| VAF(VA) | | Forward Early voltage | Volts | ∞ | 100 |
| IKF(IK) | | Corner for forward beta high-current roll-off | Amps | ∞ | 10M |
| ISE(C2) | | Base-emitter leakage saturation current | Amps | 0 | 1000 |
| NE | | Base-emitter leakage emission coefficient | | 1.5 | 2 |
| BR | | Ideal maximum reverse beta | | 1 | 0.1 |
| NR | | Reverse current emission coefficient | | 1 | |
| VAR(VB) | | Reverse Early voltage | Volts | ∞ | 100 |
| IKR | * | Corner for reverse beta high-current roll-off | Amps | ∞ | 100M |
| ISC(C4) | | Base-collector leakage saturation current | Amps | 0 | 1 |
| NC | | Base-collector leakage emission coefficient | | 2 | 2 |
| RB | * | Zero-bias (maximum) base resistance | Ohms | 0 | 100 |
| RBM | | Minimum base resistance | Ohms | RB | 100 |
| IRB | | Current at which RB falls halfway to RBM | Amps | ∞ | |
| RE | * | Emitter ohmic resistance | Ohms | 0 | 1 |
| RC | * | Collector ohmic resistance | Ohms | 0 | 10 |
| CJE | * | Base-emitter zero-bias *p-n* capacitance | Farads | 0 | 2P |
| VJE(PE) | | Base-emitter built-in potential | Volts | 0.75 | 0.75 |
| MJE(ME) | | Base-emitter *p-n* grading factor | 0.33 | 0.33 | |
| CJC | * | Base-collector zero-bias *p-n* capacitance | Farads | 0 | 1P |
| VJC(PC) | | Base-collector built-in potential | Volts | 0.75 | 0.75 |
| MJC(MC) | | Base-collector *p-n* grading factor | 0.33 | 0.33 | |
| XCJC | | Fraction of $C_{bc}$ connected internal to $R_B$ | | 1 | |
| CJS(CCS) | | Collector-substrate zero-bias *p-n* capacitance | Farads | 0 | 2PF |

**TABLE 8-1** CONTINUED

| Name | Area | Model parameters | Units | Default | Typical |
|---|---|---|---|---|---|
| VJS(PS) | | Collector-substrate built-in potential | Volts | 0.75 | |
| MJS(MS) | | Collector-substrate *p-n* grading factor | | 0 | |
| FC | | Forward-bias depletion capacitor coefficient | | 0.5 | |
| TF | | Ideal forward transit time | Seconds | 0 | 0.1NS |
| XTF | | Transit-time bias dependence coefficient | | 0 | |
| VTF | | Transit-time dependency on $V_{bc}$ | Volts | ∞ | |
| ITF | | Transit-time dependency on $I_c$ | Amps | 0 | |
| PTF | | Excess phase at 1/(2$\pi$*TF)Hz | Degrees | 0 | 30° |
| TR | | Ideal reverse transit time | Seconds | 0 | 10NS |
| EG | | Bandgap voltage (barrier height) | Electron-volts | 1.11 | 1.11 |
| XTB | | Forward and reverse beta temperature coefficient | | 0 | |
| XTI(PT) | | IS temperature-effect exponent | | 3 | |
| KF | | Flicker noise coefficient | | 0 | 6.6E−16 |
| AF | | Flicker noise exponent | | 1 | 1 |




(a) PSpice model

(b) Large-signal model

**Figure 8-4** Ebers-Moll BJT model.

As with diodes, an *area factor* is used to determine the number of equivalent parallel BJTs of a specified model. The model parameters that are affected by the area factor are marked by an asterisk (*) in Table 8-1. RC, RE, and RB represent the contact and bulk resistances per unit area of the collector, emitter, and base, respectively. The bipolar transistor is modeled as an intrinsic device. The area value, which is the relative device area, is specified in the .MODEL statement (Section 8.3), and changes the actual resistance values. The area has a default value of 1.

Some parameters have alternate names, such as VAF and VA. One can use either name, VAF or VA. These are indicated by parentheses in Table 8-1.

The parameters ISE (C2) and ISC (C4) can be either greater than or less than 1. If they are less than 1, they represent the absolute currents. If they are greater than 1, they act as the multipliers of IS instead of absolute currents. That is, the value of ISE becomes ISE*IS for ISE > 1 and that of ISC becomes ISC*IS for ISC > 1.

The dc model is defined (1) by parameters BF, C2, IK, and NE, which determine the forward-current gain, (2) by BR, C4, IKR, and VC, which determine the reverse-current gain characteristics, (3) by VA and VB, which determine the output conductance for forward and reverse regions, and (4) by the reverse saturation current IS.

Base-charge storage is modeled (1) by forward and reverse transit times TF and TF, and nonlinear depletion-layer capacitances, which are determined by CJE, PE, and ME for a b-e junction, and (2) by CJC, PC, and MC for a b-c junction. CCS is a constant collector-substrate capacitance.

The temperature dependence of the saturation current is determined by the energy gap EG and the saturation-current temperature exponent PT.

## 8-3 BJT STATEMENTS

The symbol for a bipolar junction transistor (BJT) is *Q*. The name of a bipolar transistor must start with *Q*, and it takes the general form

```
Q⟨name⟩  NC  NB  NE  NS  QNAME [(area) value]
```

where NC, NB, NE, and NS are the collector, base, emitter, and substrate nodes, respectively. QNAME could be any name of up to eight characters. The substrate node is optional; if not specified, it defaults to ground. Positive current is the current that flows into a terminal. That is, the current flows from the collector node through the device to the emitter node for an NPN BJT.

### *Some Statements for BJTs*

```
QIN   5  7  8   2N2222
Q5    2  4  5   2N2907   1.5
QX    1  4  9   NMOD
.MODEL 2N2222 NPN(IS=3.108E-15 XTI=3 EG=1.11 VAF=131.5 BF=217.5
```

```
+  NE=1.541 ISE=190.7E-15 IKF=1.296 XTB=1.5 BR=6.18 NC=2 ISC=0 IKR=0
+  RC=1 CJC=14.57E-12 VJC=.75 MJC=.3333 FC=.5 CJE=26.08E-12 VJE=.75
+  MJE=.3333 TR=51.35E-9 TF=451E-12 ITF=.1 VTF=10 XTF=2)
.MODEL 2N2907 PNP(IS=9.913E-15 XTI=3 EG=1.11 VAF=90.7 BF=197.8
+  NE=2.264 ISE=6.191E-12 IKF=.7322 XTB=1.5 BR=3.369 NC=2 ISC=0 IKR=0
+  RC=1 CJC=14.57E-12 VJC=.75 MJC=.3333 FC=.5 CJE=20.16E-12 VJE=.75
+  MJE=.3333 TR=29.17E-9 TF=405.7E-12 ITF=.4 VTF=10 XTF=2)
.MODEL NMOD NPN
```

*Note.* A + (plus) sign at the first column indicates the continuation of the statement preceding it.

## 8-4 BJT PARAMETERS

The data sheet for NPN transistor of type 2N2222A is shown in Fig. 8-5. SPICE parameters are not quoted directly in the data sheet. Some versions of SPICE (e.g., PSpice) support device library files that give the model parameters. The library file EVAL.LIB contains the list of devices and their model statements in the student's version of PSpice. The software PARTS of PSpice can generate SPICE models from the data-sheet parameters of diodes. SPICE model parameters are also supplied by some manufacturers. However, some parameters that significantly influence the performance of a transistor can be determined from the data sheet [7, 8].

The diode characteristic described by Eq. (7-1) can be applied to an NPN transistor by selecting appropriate subscripts, as in the following equation:

$$I_C = I_S[e^{V_{BE}/\eta V_T} - 1] \tag{8-1}$$

Using Eq. (7-8), the difference in the base-emitter voltages can be expressed by

$$V_{BE2} - V_{BE1} = 2.3\ \eta V_T \log\left(\frac{I_{C2}}{I_{C1}}\right) \tag{8-2}$$

From the data sheet for $V_{BD}$ versus $I_C$, we get $V_{BE1} = 0.6$ V at $I_{C1} = 0.2$ mA, and $V_{BE2} = 0.7$ V at $I_{C2} = 20$ mA. Assuming $V_T = 25.8$ mV $= 0.0258$, we can apply Eq. (8-2) to find the *emission coefficient* $\eta$ as follows:

$$0.7 - 0.6 = 2.3\ \eta \times 0.0258 \log\left(\frac{20}{0.2}\right)$$

which gives $\eta = 0.843$. Since we did not include any contact and bulk resistance RE of the emitter, we got the value $\eta < 1$; its practical value is $\eta \geq 1$. Let us assume $\eta = 1$. For $\eta = 1$, and $V_{BE2} = 0.7$ V at $I_{C2} = 20$ mA, we can apply Eq. (8-1) to find the saturation current $I_S$

$$20 \text{ mA} = I_S[e^{0.7/(1 \times 0.0258)} - 1]$$

which gives $I_S = 3.295\text{E}-14$ A. The dc current gain at 150 mA is $h_{FE} = 100$ to 300. Taking the geometric mean gives $BF = \sqrt{(100 \times 300)} \approx 173$.

National Semiconductor

# 2N2222 2N2222A | PN2222 PN2222A | MMBT2222 MMBT2222A | MPQ2222*



TO-18

TL/G/10100-9




TO-92

TL/G/10100-1




TO-236
(SOT-23)

TL/G/10100-5




TO-116

TL/G/10100-7

## NPN General Purpose Amplifier

## Electrical Characteristics $T_A = 25°C$ unless otherwise noted

| Symbol | Parameter | | Min | Max | Units |
|---|---|---|---|---|---|
| **OFF CHARACTERISTICS** | | | | | |
| $V_{(BR)CEO}$ | Collector-Emitter Breakdown Voltage (Note 1) ($I_C = 10$ mA, $I_B = 0$) | 2222<br>2222A | 30<br>40 | | V |
| $V_{(BR)CBO}$ | Collector-Base Breakdown Voltage ($I_C = 10$ μA, $I_E = 0$) | 2222<br>2222A | 60<br>75 | | V |
| $V_{(BR)EBO}$ | Emitter Base Breakdown Voltage ($I_E = 10$ μA, $I_C = 0$) | 2222<br>2222A | 5.0<br>6.0 | | V |
| $I_{CEX}$ | Collector Cutoff Current ($V_{CE} = 60V$, $V_{EB(OFF)} = 3.0V$) | 2222A | | 10 | nA |
| $I_{CBO}$ | Collector Cutoff Current<br>($V_{CB} = 50V$, $I_E = 0$)<br>($V_{CB} = 60V$, $I_E = 0$)<br>($V_{CB} = 50V$, $I_E = 0$, $T_A = 150°C$)<br>($V_{CB} = 60V$, $I_E = 0$, $T_A = 150°C$) | <br>2222<br>2222A<br>222<br>2222A | | <br>0.01<br>0.01<br>10<br>10 | μA |
| $I_{EBO}$ | Emitter Cutoff Current ($V_{EB} = 3.0V$, $I_C = 0$) | 2222A | | 10 | nA |
| $I_{BL}$ | Base Cutoff Current ($V_{CE} = 60V$, $V_{EB(OFF)} = 3.0$) | 2222A | | 20 | nA |
| **ON CHARACTERISTICS** | | | | | |
| $h_{FE}$ | DC Current Gain<br>($I_C = 0.1$ mA, $V_{CE} = 10V$)<br>($I_C = 1.0$ mA, $V_{CE} = 10V$)<br>($I_C = 10$ mA, $V_{CE} = 10V$)<br>($I_C = 10$ mA, $V_{CE} = 10V$, $T_A = -55°C$)<br>($I_C = 150$ mA, $V_{CE} = 10V$) (Note 1)<br>($I_C = 150$ mA, $V_{CE} = 1.0V$) (Note 1)<br>($I_C = 500$ mA, $V_{CE} = 10V$) (Note 1)<br> | <br><br><br><br><br><br><br>2222<br>2222A | <br>35<br>50<br>75<br>35<br>100<br>50<br>30<br>40 | <br><br><br><br><br>300 | |

Note 1: Pulse Test: Pulse Width ≤ 300 μs, Duty Cycle ≤ 2.0%.

*16-SOIC version also available. Contact factory.

**Figure 8-5** Data sheet for transistor of type 2N2222A (Courtesy of National Semiconductor, Inc.).

# NPN General Purpose Amplifier (Continued)

## Electrical Characteristics $T_A$ = 25°C unless otherwise noted (Continued)

| Symbol | Parameter | | Min | Max | Units |
|---|---|---|---|---|---|
| **ON CHARACTERISTICS (Continued)** | | | | | |
| $V_{CE(sat)}$ | Collector-Emitter Saturation Voltage (Note 1) | | | | V |
| | ($I_C$ = 150 mA, $I_B$ = 15 mA) | 2222 | | 0.4 | |
| | | 2222A | | 0.3 | |
| | ($I_C$ = 500 mA, $I_B$ = 50 mA) | 2222 | | 1.6 | |
| | | 2222A | | 1.0 | |
| $V_{BE(sat)}$ | Base-Emitter Saturation Voltage (Note 1) | | | | V |
| | ($I_C$ = 150 mA, $I_B$ = 15 mA) | 2222 | 0.6 | 1.3 | |
| | | 2222A | 0.6 | 1.2 | |
| | ($I_C$ = 500 mA, $I_B$ = 50 mA) | 2222 | | 2.6 | |
| | | 2222A | | 2.0 | |
| **SMALL-SIGNAL CHARACTERISTICS** | | | | | |
| $f_T$ | Current Gain—Bandwidth Product (Note 3) | | | | MHz |
| | ($I_C$ = 20 mA, $V_{CE}$ = 20V, f = 100 MHz) | 2222 | 250 | | |
| | | 2222A | 300 | | |
| $C_{obo}$ | Output Capacitance (Note 3) ($V_{CB}$ = 10V, $I_E$ = 0, f = 100 kHz) | | | 8.0 | pF |
| $C_{ibo}$ | Input Capacitance (Note 3) | | | | pF |
| | ($V_{EB}$ = 0.5V, $I_C$ = 0, f = 100 kHz) | 2222 | | 30 | |
| | | 2222A | | 25 | |
| $r_{b'}C_C$ | Collector Base Time Constant ($I_E$ = 20 mA, $V_{CB}$ = 20V, f = 31.8 MHz) | 2222A | | 150 | ps |
| NF | Noise Figure ($I_C$ = 100 μA, $V_{CE}$ = 10V, $R_S$ = 1.0 kΩ, f = 1.0 kHz) | 2222A | | 4.0 | dB |
| $Re(h_{ie})$ | Real Part of Common-Emitter High Frequency Input Impedance ($I_C$ = 20 mA, $V_{CE}$ = 20V, f = 300 MHz) | | | 60 | Ω |

**SWITCHING CHARACTERISTICS**

| Symbol | Parameter | Conditions | | Min | Max | Units |
|---|---|---|---|---|---|---|
| $t_D$ | Delay Time | ($V_{CC}$ = 30V, $V_{BE(OFF)}$ = 0.5V, $I_C$ = 150 mA, $I_{B1}$ = 15 mA) | except MPQ2222 | | 10 | ns |
| $t_R$ | Rise Time | | | | 25 | ns |
| $t_S$ | Storage Time | ($V_{CC}$ = 30V, $I_C$ = 150 mA, $I_{B1}$ = $I_{B2}$ = 15 mA | except MPQ2222 | | 225 | ns |
| $t_F$ | Fall Time | | | | 60 | ns |

**Note 1:** Pulse Test: Pulse Width < 300 μs, Duty Cycle ≤ 2.0%.

**Note 2:** For characteristics curves, see Process 19.

**Note 3:** $f_T$ is defined as the frequency at which $|h_{fe}|$ extrapolates to unity.

**Note 4:** 2N also available in JAN/TX/V series.

**Figure 8-5** *Continued*

# Process 19

**SMALL SIGNAL CHARACTERISTICS (f = 1.0 kHz)**

| Symbol | Parameter | Conditions | Typ | Units |
|---|---|---|---|---|
| $h_{ie}$ | Input Resistance | $I_C$ = 10 mA, $V_{CE}$ = 10V | 700 | Ω |
| $h_{oe}$ | Output Conductance | $I_C$ = 10 mA, $V_{CE}$ = 10V | 120 | μmhos |
| $h_{fe}$ | Small Signal Current Gain | $I_C$ = 10 mA, $V_{CE}$ = 10V | 240 | |
| $h_{re}$ | Voltage Feedback Ratio | $I_C$ = 10 mA, $V_{CE}$ = 10V | 460 | $\times 10^{-6}$ |

**TYPICAL COMMON EMITTER CHARACTERISTICS (f = 1.0 kHz)**










TL/G/10034-24

**DC Current Gain vs Collector Current**




**Base-Emitter ON Voltage vs Collector Current**




**Collector Reverse Current vs Ambient Temperature**




**Emitter Cutoff Current vs Ambient Temperature**




**Maximum Power Dissipation vs Ambient Temperature**




**Maximum Power Dissipation vs Case Temperature**




TL/G/10034-22

**Figure 8-5** *Continued*

# Process 19

### Emitter Transition and Output Capacitance vs Reverse Bias Voltage




### Contours of Constant Gain Bandwidth Product ($f_T$)




### Noise Figure vs Collector Current




### Noise Figure vs Frequency




### Collector Saturation Voltage vs Collector Current




### Base Saturation Voltage vs Collector Current




### Switching Time vs Collector Current




### Turn On and Turn Off Times vs Collector Current




TL/G/10034-23

**Figure 8-5** *Continued*

The input capacitance at the base-emitter junction is $C_{ibo} = 25$ pF at $V_{EB} = 0.5$ V, $I_C = 0$ (reverse-biased). Since $C_{je} = C_{ibo}$, then $C_{jeo}$ can be found from

$$C_{je} = \frac{C_{jeo}}{(1 + V_{EB}/V_{je})^{M_{je}}} \tag{8-3}$$

where $M_{je}$ = MJE ≈ 1/3, and $V_{je}$ = VJE ≈ 0.75 V. Equation (8-3) gives $C_{jeo}$ = CJE = 29.6 pF at $V_{BE} = 0$ V.

The output capacitance is $C_{obo} = 8$ pF at $V_{CB} = 10$ V, $I_E = 0$ (reverse-biased). Since $C_\mu = C_{obo}$, then $C_{\mu o}$ can be found from

$$C_\mu = \frac{C_{\mu o}}{(1 + V_{CB}/V_{jc})^{M_{jc}}} \tag{8-4}$$

where $M_{jc}$ = MJC ≈ 1/3, and $V_{jc}$ ≈ VJC = 0.75 V. From Eq. (8-4), $C_{\mu o}$ = CJC = 19.4 pF at $V_{CB} = 0$ V.

The transition frequency $f_{T(\min)} = 300$ MHz at $V_{CE} = 20$ V, $I_C = 20$ mA. The transition period is $\tau_T = 1/2\pi f_T = 1/(2\pi \times 300 \text{ MHz}) = 530.5$ ps. Assuming $V_{BE} = 0.7$ V, $V_{CB} \approx V_{CE} - V_{BE} = 20 - 0.7 = 19.3$ V, and Eq. (8-4) gives $C_\mu = 6.49$ pF.

Since the transition frequency $f_{T(\min)} = 300$ MHz is specified at $I_C = 20$ mA, we need to find the transconductance $g_m$ (at $I_C = 20$ mA), which is given by

$$g_m = \frac{I_C}{V_T} \tag{8-5}$$

$$= 20 \text{ mA}/25.8 \text{ mV} = 775.2 \text{ mA/V}$$

The transition period $\tau_T$ is related to forward transit time $\tau_F$ by

$$\tau_T = \tau_F + \frac{C_{je}}{g_m} + \frac{C_\mu}{g_m} \tag{8-6}$$

or

$$530.5 \text{ ps} = \tau_F + \frac{25 \text{ pF}}{0.7752} + \frac{6.49 \text{ pF}}{0.7752}$$

which gives $\tau_F$ = TF = 489.88 ps.

The output conductance $h_{oe}$ of a transistor is related to the collector current $I_C$ and the *Early voltage* $V_A$ by

$$\frac{1}{h_{oe}} = \frac{V_A}{I_C} \tag{8-7}$$

From the data sheet, $h_{oe} = 120$ μmhos at $I_C = 10$ mA and $V_{CE} = 10$ V. From Eq. (8-7), the Early voltage $V_A$ becomes

$$V_A = \text{VA} = I_C/h_{oe} = 10 \text{ mA}/120 \text{ μmhos} = 83.3 \text{ V}.$$

The reverse transit time can be approximated to $\tau_R$ = TR = 10 $\tau_F$ = 4.9 ns.

The model statement for transistor 2N2222A is

```
.MODEL  Q2N2222A NPN (IS=3.295E-14 BF=173 VA=83.3V CJE=29.6PF CJC=19.4PF
+        TF=489.88PS TR=4.9NS)
```

This model is used to plot the characteristics of the BJT as illustrated in Example 8-1. It may be necessary to modify the parameter values to conform to the actual characteristics.

*Note.* If a model parameter is not specified in the model statement, SPICE assumes its default value. However, it should be noted that some default values represent ideal conditions (e.g., TR = 0 and VAF = ∞). More accurate results can be obtained by using the typical values (e.g., TR = 0.1NS and VAF = 100V) rather than the default ones.

## 8-5 EXAMPLES OF BJT CIRCUITS

The PSpice simulation of BJT circuits requires specifying the BJT model parameters. If the model parameters are not specified, PSpice assumes the default values that are given in Table 8-1. The following examples illustrate the PSpice simulation of BJT circuits.

### Example 8-1

For the NPN BJT transistor of Fig. 8-6, plot the output characteristics ($I_C$ versus $V_{CE}$) if $V_{CE}$ is varied from 0 to 10 V in steps of 0.02 V and $I_B$ is varied from 0 to 1 mA in steps of 200 $\mu$A. Use the model parameters that were determined in Section 8-4. Print the details of the small-signal parameters at the operating point for $I_B$ = 1 mA and $V_{CE}$ = 12 V.




**Figure 8-6** A circuit with an NPN BJT.

**Solution** The listing of the circuit file follows.

**Example 8-1 NPN BJT characteristics**

```
▲ IB    0  1  DC  1MA               ; Base current
  VCE   2  0  DC  12V               ; Collector-emitter voltage
▲▲ Q1    2   1   0    Q2N2222A          ; BJT statement
   .MODEL  Q2N2222A  NPN (IS=2.105E-16 BF=173 VA=83.3V CJE=29.6PF CJC=19.4PF
   +        TF=489.88PS TR=4.9NS)       ; Model parameters
▲▲▲ .DC  VCE  0  10V  0.02V  IB  0  1MA  200UA ; Dc sweep for VCE and IB
     .PROBE                               ; Graphics post-processor
.END
```

The output characteristics, which are plots of $I_C$ versus $V_{CE}$, are shown in Fig. 8-7. The students are encouraged to compare the transistor characteristics with those obtained by using the model parameters of transistor 2N2222A, which is listed




**Figure 8-7** Output characteristics of the BJT in Example 8.1.

in the PSpice library file EVAL.LIB. This can be done by replacing the .MODEL statement in the above circuit file by the .LIB EVAL.LIB statement.

The small-signal parameters of the transistor at the operating point, which are obtained from the output file EX8.1.OUT, are as follows.

```
****     OPERATING POINT INFORMATION          TEMPERATURE =  27.000 DEG C
     NAME          Q1
     MODEL         Q2N2222A
     IB            1.00E-03
     IC            1.96E-01
     VBE           8.88E-01
     VBC          -1.11E+01
     VCE           1.20E+01
     BETADC        1.96E+02
     GM            7.58E+00
     RPI           2.59E+01
     RX            0.00E+00
     RO            4.82E+02
     CBE           3.77E-09
     CBC           7.80E-12
     CBX           0.00E+00
     CJS           0.00E+00
     BETAAC        1.96E+02
     FT            3.19E+08
```

### Example 8-2

A bipolar transistor circuit is shown in Fig. 8-8(a), where the output is taken from node 4. Calculate and print the sensitivity of the collector current with respect to all parameters. Print the details of the bias point. The equivalent circuit for transistor $Q_1$ is shown in Fig. 8-8(b).




**Figure 8-8** Bipolar transistor circuit.

**Solution** The listing of the circuit file follows.

**Example 8-2 Biasing sensitivity of bipolar transistor amplifier**

```
▲ *  Supply voltage is 15 V DC.
  VCC  7 0  DC 15V
  *  A dummy voltage source of 0 V to measure the collector current
  VRC  6  4  DC  0V
▲▲ R1  7  3  47K
   R2  3  0  2K
   RC  7  6  10K
   RE   5  0  2K
   *   Subcircuit call for transistor model QMOD, and the substrate is
   *   connected to ground by default.
   XQ1 4  3  5  QMOD
   *   Subcircuit definition for QMOD
   .SUBCKT  QMOD  6  7  5
   RB  1  2  100
   RE  3  5  1
   RC  4  6  10
   RBE 2  3  1K
   RO  4  3  100K
```

```
  *   A dummy voltage source of 0 V to measure the controlling current
  VI  7  1  DC 0V
  F1  4  3  VI  20
  *   End of subcircuit definition
  .ENDS  QMOD
▲▲▲ .OPTIONS NOPAGE NOECHO
    *   Sensitivity of collector current (which is the current through voltage
    *   source VRC)
    .SENS  I(VRC)
.END
```

The .SENS command does not require a .PRINT command for printing the output. The output for the sensitivity analysis and the bias point follow.

```
 ****      SMALL-SIGNAL BIAS SOLUTION                    TEMPERATURE =   27.000 DEG C
 NODE      VOLTAGE     NODE      VOLTAGE      NODE     VOLTAGE      NODE     VOLTAGE

(    3)      .5960    (    4)   12.1520    (    5)     .5864    (    6)   12.1520
(    7)    15.0000    (XQ1.1)     .5960    (XQ1.2)     .5952    (XQ1.3)     .5867
(XQ1.4)    12.1500

       VOLTAGE SOURCE CURRENTS
        NAME          CURRENT

       VCC          -5.912E-04
       VRC           2.848E-04
       XQ1.VI        8.456E-06
       TOTAL POWER DISSIPATION   8.87E-03 WATTS

 ****      DC SENSITIVITY ANALYSIS                       TEMPERATURE =   27.000 DEG C
DC SENSITIVITIES OF OUTPUT I(VRC)
                                      ELEMENT          NORMALIZED
       ELEMENT        ELEMENT       SENSITIVITY       SENSITIVITY
        NAME           VALUE        (AMPS/UNIT)     (AMPS/PERCENT)

       R1           4.700E+04      -5.481E-09        -2.576E-06
       R2           2.000E+03       1.252E-07         2.505E-06
       RC           1.000E+04      -3.134E-10        -3.134E-08
       RE           2.000E+03      -1.288E-07        -2.576E-06
       XQ1.RB       1.000E+02      -3.705E-09        -3.705E-09
       XQ1.RE       1.000E+00      -1.288E-07        -1.288E-09
       XQ1.RC       1.000E+01      -3.134E-10        -3.134E-11
       XQ1.RBE      1.000E+03      -3.705E-09        -3.705E-08
       XQ1.RO       1.000E+05      -1.273E-10        -1.273E-07
       VCC          1.500E+01       1.898E-05         2.848E-06
       VRC          0.000E+00      -1.101E-06         0.000E+00
       XQ1.VI       0.000E+00      -4.381E-04         0.000E+00
       JOB CONCLUDED
       TOTAL JOB TIME          2.52
```

## Example 8-3

A bipolar Darlington pair amplifier is shown in Fig. 8-9. Calculate and print the voltage gain, the input resistance, and the output resistance. The input voltage is 5 V. The model parameters of the bipolar transistors are BF=100, BR=1, RB=5, RC=1, RE=0, VJE=0.8, and VA=100.




**Figure 8-9** Darlington pair amplifier.

**Solution** The listing of the circuit file follows.

**Example 8-3 Darlington pair**

```
▲ .OPTIONS  NOPAGE  NOECHO
  VCC  2  0  DC  12V
  VIN  1  0  DC  5V
▲▲ *   BJTs with model QM
    Q1   2  1  3  QM
    Q2   2  3  4  QM
    RB   2  1  47K
    RE   4  0  4.7K
    *    Model QM for NPN BJTs
    .MODEL QM  NPN (BF=100 BR=1 RB=5 RC=1 RE=0 VJE=0.8 VA=100)
▲▲▲ *   Transfer-function analysis to calculate dc gain, input
      *   resistance, and output resistance
      .TF  V(4)  VIN
.END
```

The results of the transfer-function analysis follow.

```
****     SMALL-SIGNAL BIAS SOLUTION            TEMPERATURE =   27.000 DEG C
 NODE     VOLTAGE     NODE    VOLTAGE      NODE    VOLTAGE     NODE     VOLTAGE

(    1)    5.0000    (    2)   12.0000    (    3)    4.3560    (    4)    3.5909

      VOLTAGE SOURCE CURRENTS
      NAME          CURRENT

      VCC         -9.129E-04
      VIN          1.489E-04
      TOTAL POWER DISSIPATION   1.02E-02  WATTS
```

```
****     SMALL-SIGNAL CHARACTERISTICS
      V(4)/VIN =   9.851E-01
      INPUT RESISTANCE AT VIN =   4.696E+04
      OUTPUT RESISTANCE AT V(4)  =   6.679E+01
          JOB CONCLUDED
          TOTAL JOB TIME              2.97
```

## Example 8-4

A bipolar transistor amplifier circuit is shown in Fig. 8-10. The output is taken from node 6. Calculate and plot the magnitude and phase of the voltage gain for frequencies from 1 Hz to 10 kHz with a decade increment and with 10 points per decade. The input voltage for ac analysis is 10 mV. Calculate and plot the transient response of voltages at nodes 4 and 6 for an input voltage of $v_{in} = 0.01 \sin(2\pi \times 1000t)$ and for a duration of 0 to 2 ms in steps of 50 $\mu$s. The details of ac and transient analysis operating points should be printed. The model parameters of the PNP BJT are IS=2E−16, BF=50, BR=1, RB=5, RC=1, RE=0, TF=0.2NS, TR=5NS, CJE=0.4PF, VJE=0.8, ME=0.4, CJC=0.5PF, VJC=0.8, CCS=1PF, and VA=100.




**Figure 8-10** Bipolar transistor amplifier circuit.

**Solution** The listing of the circuit file follows.

**Example 8-4 Bipolar transistor amplifier**

```
▲ *  Input voltage is 10 mV peak for ac analysis and for transient response:
  *  It is 10 mV peak at 1 kHz with zero-offset value.
  VIN  1  0  AC  10MV  SIN(0  10MV  1KHZ)
  VCC  0   7  DC  15V
▲▲ RS  1  2  500
   R1  7  3  47K
   R2  3  0  5K
   RC  7  4  10K
   RE  5  0  2K
   RL  6  0  20K
   C1  2  3  1UF
   C2  4  6  1UF
   CE  5  0  10UF
```

```
   *  Transistor Q1 with model QM
   Q1  4  3  5  0  QM
   * Model QM for PNP transistors
   .MODEL  QM  PNP (IS=2E-16 BF=50 BR=1 RB=5 RC=1 RE=0 TF=0.2NS TR=5NS
   +         CJE=0.4PF VJE=0.8 ME=0.4 CJC=0.5PF VJC=0.8 CCS=1PF VA=100)
▲▲▲ *   Plot the results of transient analysis for voltages at nodes 4, 6, and 1
    .PLOT  TRAN  V(4) V(6) V(1)
    *   Plot the results of the ac analysis for the magnitude and phase angle
    *   of output voltage at node 6.
    .PLOT  AC  VM(6)  VP(6)
    .OPTIONS  NOPAGE NOECHO
    *   Transient analysis for 0 to 2 ms with 50-μs increment
    *   Print details of transient analysis operating point.
    .TRAN/OP  50US  2MS
    *  AC analysis from 1 Hz to 10 KHz with a decade increment and 10 points
    *  per decade
    .AC  DEC  10  1HZ  10KHZ
    *  Print the details of the ac analysis operating point.
    .OP
    .PROBE
.END
```

*Note.* .PLOT statements generate graphical plots in the output file. If the .PROBE command is included, there is no need for the .PLOT command.

The determination of the operating point is the first step in analyzing a circuit with nonlinear devices (e.g., bipolar transistors). The equivalent circuit for determining the ac analysis (or dc analysis) bias point of the amplifier in Fig. 8-10 is shown in Fig. 8-11, where the capacitors are open-circuited. The details of the bias point follow.




**Figure 8-11** Equivalent circuit for dc bias calculation.

```
****     SMALL-SIGNAL BIAS SOLUTION              TEMPERATURE =   27.000 DEG C
 NODE    VOLTAGE    NODE    VOLTAGE     NODE    VOLTAGE    NODE     VOLTAGE

(    1)   0.0000   (    2)   0.0000   (    3)   -1.4280   (    4)  -11.5240
(    5)   -.7016   (    6)   0.0000   (    7)  -15.0000
```

```
VOLTAGE SOURCE CURRENTS
NAME         CURRENT

VIN          0.000E+00
VCC         -6.364E-04
TOTAL POWER DISSIPATION   9.55E-03  WATTS
```

Once the dc bias point is determined, PSpice generates a small-signal model of the BJT. This model is similar to that in Fig. 8-12. PSpice replaces the transistor with this circuit model. It should be noted that this model is valid only at the operating point. The details of the operating point and model values follow.




**Figure 8-12** Small-signal equivalent circuit of bipolar transistors.

```
****      OPERATING POINT INFORMATION             TEMPERATURE =   27.000 DEG C
**** BIPOLAR JUNCTION TRANSISTORS
NAME          Q1
MODEL         QM
IB           -3.16E-06
IC           -3.48E-04
VBE          -7.26E-01
VBC           1.01E+01
VCE          -1.08E+01
BETADC        1.10E+02
GM            1.34E-02
RPI           8.19E+03
RX            5.00E+00
RO            3.17E+05
CBE           3.39E-12
CBC           2.11E-13
CBX           0.00E+00
CJS           1.00E-12
BETAAC        1.10E+02
FT            5.94E+08
```

Prior to the transient analysis, PSpice determines the small-signal parameters of the nonlinear devices and the potentials of the various nodes. The method for the calculation of the transient analysis bias point differs from that of the dc analysis bias point because, in transient analysis, all the nodes have to be assigned initial values, and the nonlinear sources may have transient values at the beginning of transient analysis. The capacitors, which may have initial values, therefore remain as parts of the circuit. The equivalent circuit for determining the transient analysis bias point for the circuit in Fig. 8-10 is shown in Fig. 8-13. Since the capacitors in Fig. 8-10 do not have any initial values, the bias points for dc and transient analysis are the same. There, the small-signal parameters are also the same. The details of the transient analysis bias point and the small-signal parameters are given next to compare with those of dc analysis.




**Figure 8-13** Equivalent circuit for the transient analysis bias point.

```
****      INITIAL TRANSIENT SOLUTION              TEMPERATURE =   27.000 DEG C
 NODE     VOLTAGE      NODE     VOLTAGE     NODE     VOLTAGE      NODE     VOLTAGE

(   1)    0.0000     (    2)    0.0000     (    3)    -1.4280    (    4)   -11.5240
(   5)    -.7016     (    6)    0.0000     (    7)   -15.0000

      VOLTAGE SOURCE CURRENTS
      NAME          CURRENT

      VIN          0.000E+00
      VCC         -6.364E-04

      TOTAL POWER DISSIPATION    9.55E-03   WATTS

****      OPERATING POINT INFORMATION             TEMPERATURE =   27.000 DEG C
**** BIPOLAR JUNCTION TRANSISTORS

NAME          Q1
MODEL         QM
IB            -3.16E-06
IC            -3.48E-04
```

| | |
|---|---|
| VBE | -7.26E-01 |
| VBC | 1.01E+01 |
| VCE | -1.08E+01 |
| BETADC | 1.10E+02 |
| GM | 1.34E-02 |
| RPI | 8.19E+03 |
| RX | 5.00E+00 |
| RO | 3.17E+05 |
| CBE | 3.39E-12 |
| CBC | 2.11E-13 |
| CBX | 0.00E+00 |
| CJS | 1.00E-12 |
| BETAAC | 1.10E+02 |
| FT | 5.94E+08 |

The frequency and transient responses are shown in Figs. 8-14 and 8-15, respectively.




**Figure 8-14** Frequency response for Example 8-4.




**Figure 8-15** Transient response for Example 8-4.

## Example 8-5

If the transistor in Fig. 8-10 is replaced by the equivalent circuit of Fig. 8-16, repeat Example 8-4. There is no need to print the details of the operating point.




**Figure 8-16** Subcircuit for PNP bipolar transistor.

**Solution** The listing of the circuit file follows.

**Example 8-5 Bipolar transistor amplifier**

```
▲ * Input voltage is 10 mV peak for ac analysis and for transient response:
```

```
*  It is 10 mV peak at 1 kHz with zero-offset value.
VIN  1  0  AC  1  SIN(0 0.01 1KHZ)
VCC  0  7  DC  15V
▲▲ RS  1  2  500
   R1  7  3  47K
   R2  3  0  2K
   RC  7  4  10K
   RE  5  0  2K
   RL  6  0  20K
   C1  2  3  1UF
   C2  4  6  1UF
   CE  5  0  10UF
   *  Calling subcircuit for transistor model TRANS
   XQ1  4  3  5  TRANS
   *  Subcircuit definition for TRANS
   .SUBCKT  TRANS  6  7  5
   RB  1  2  100
   RE  3  5  1
   RC  4  6  10
   RPI 2  3  1K
   CPI 2  3  2PF
   RU  2  4  1MEG
   CU  2  4  1PF
   RO  4  3  100K
   CCS 6  0  2PF
   *  A dummy voltage source of 0 V through which the controlling current flows
   VI  1  7  DC 0V
   * The collector current is controlled by the current through source VI.
   F1  3  4  VI  20
   *  End of subcircuit definition
   .ENDS  TRANS
▲▲▲ .OPTIONS  NOPAGE  NOECHO
      *   Transient analysis for 0 to 2 ms with 50-μs increment
      .TRAN  50US  2MS
      *  Ac analysis from 1 Hz to 10 KHz with a decade increment and
      *  10 points per decade
      .AC  DEC  10  1HZ  10KHZ
      *   Plot the results of transient analysis for voltages at nodes 4, 6, and 1.
      .PLOT  TRANS  V(4)  V(6)  V(1)
      *   Plot the results of ac analysis for the magnitude and phase angle
      *   of voltage at node 6.
      .PLOT  AC  VM(6)  VP(6)
      .PROBE
.END
```

The frequency and transient responses are shown in Figs. 8-17 and 8-18, respectively. The .PLOT statements generate graphical plots in the output file. If the .PROBE command is included, there is no need for the .PLOT command.




**Figure 8-17** Frequency response for Example 8-5.




**Figure 8-18** Transient response for Example 8-5.

## Example 8-6

A two-stage bipolar transistor amplifier is shown in Fig. 8-19. The output is taken from node 9. Plot (a) the magnitude and phase angle of the voltage gain and (b) the magnitude of input impedance for frequencies from 10 Hz to 10 MHZ with a decade increment and 10 points per decade. The peak input voltage is 1 mV. The model parameters of the BJTs are IS=2E−16, BF=50, BR=1, RB=5, RC=1, RE=0, CJE=0.4PF, VJE=0.8, ME=0.4, CJC=0.5PF, VJC=0.8, CCS=1PF, and VA=100.




**Figure 8-19** Two-stage BJT amplifier.

**Solution** The listing of the circuit file follows.

**Example 8-6 Two-stage BJT amplifier**

```
▲ VCC 10  0  DC  15V
  *  Input voltage is 1 mV peak for frequency response.
  VIN  1  0  AC  1MV
  *  A dummy voltage source of 0 V to measure the input current
  VX   1   12  DC  0V
▲▲ RS    12  2   150
   C1    2   3   10UF
   R1    10  3   200K
   R2    3   0   50K
   *  Transistors Q1 and Q2 have model name QM.
   Q1   4  3  5  0  QM
   Q2   7  6  8  0  QM
   RC1  10  4   12K
   RE1  5   0   3.6K
   CE1  5   0   15UF
   C2   4   6   10UF
   R3   10  6   120K
   R4   6   0   30K
   RC2  10  7   6.8K
   RE2  8   0   3.6K
   CE2  8   0   25UF
```

## Example 8-6

A two-stage bipolar transistor amplifier is shown in Fig. 8-19. The output is taken from node 9. Plot (a) the magnitude and phase angle of the voltage gain and (b) the magnitude of input impedance for frequencies from 10 Hz to 10 MHZ with a decade increment and 10 points per decade. The peak input voltage is 1 mV. The model parameters of the BJTs are IS=2E−16, BF=50, BR=1, RB=5, RC=1, RE=0, CJE=0.4PF, VJE=0.8, ME=0.4, CJC=0.5PF, VJC=0.8, CCS=1PF, and VA=100.




**Figure 8-19** Two-stage BJT amplifier.

**Solution** The listing of the circuit file follows.

**Example 8-6 Two-stage BJT amplifier**

```
▲ VCC 10   0  DC  15V
  *   Input voltage is 1 mV peak for frequency response.
  VIN  1  0  AC  1MV
  *  A dummy voltage source of 0 V to measure the input current
  VX   1    12  DC  0V
▲▲ RS    12  2    150
   C1    2   3    10UF
   R1    10  3    200K
   R2    3   0    50K
   *  Transistors Q1 and Q2 have model name QM.
   Q1   4  3  5  0  QM
   Q2   7  6  8  0  QM
   RC1  10  4    12K
   RE1  5    0    3.6K
   CE1  5    0    15UF
   C2   4    6    10UF
   R3   10   6    120K
   R4   6    0    30K
   RC2  10   7    6.8K
   RE2  8    0    3.6K
   CE2  8    0    25UF
```

```
C3   7   9   10UF
RL   9   0   10K
*  Model statement for NPN transistors whose model name is QM
.MODEL  QM  NPN (IS=2E-16 BF=50 BR=1 RB=5 RC=1 RE=0 CJE=0.4PF
+                VJE=0.8 ME=0.4 CJC=0.5PF VJC=0.8 CCS=1PF VA=100)
▲▲▲ *  Ac analysis from 10 Hz to 10 MHz with a decade increment and 10
    *  points per decade
    .AC  DEC  10  10HZ  10MEGHZ
    .PLOT  AC  VM(9)  VP(9)
    .PROBE
.END
```

The results of the frequency response are shown in Fig. 8-20. If the .PROBE command is included, there is no need for the .PLOT command.




**Figure 8-20** Frequency response for Example 8-6.

## Example 8-7

A two-stage amplifier with shunt-series feedback is shown in Fig. 8-21. Plot (a) the magnitude and phase angle of voltage gain and (b) the magnitude of the input impedance if the frequency is varied from 100 Hz to 100 MHz in decade steps with 10 points per decade. The peak input voltage is 10 mV. The model parameters of the BJTs are IS=2E−16, BF=50, BR=1, RB=5, RC=1, RE=0, CJE=0.4PF, VJE=0.8, ME=0.4, CJC=0.5PF, VJC=0.8, CCS=1PF, and VA=100.




**Figure 8-21** Two-stage BJT amplifier with shunt-series feedback.

## Solution

### Example 8-7 Two-stage BJT amplifier with shunt-series feedback

```
▲ VCC 10   0 DC 15V
  *   Input voltage of 10 mV peak for frequency response
  VIN  1  0  AC  10MV
  *   A dummy voltage source of 0 V
  VX   1   12  DC  0V
▲▲ RS   12   2  150
   C1   2    3  10UF
   R1   10   3 200K
   R2  3  0    50K
   *  Substrate of BJTs with model QM is connected to node 0.
   Q1  4  3  5  0  QM
   Q2   7  6  8  0  QM
   RC1  10  4   12K
   RE1  5   0   3.6K
   CE1  5   0   15UF
   C2   4   6   10UF
   R3   10  6   120K
   R4   6   0   30K
   RC2  10  7   6.8K
   RE2  8   0   3.6K
   CF   11  8   10UF
   RF   2   11  25K
   C3   7  9    10UF
   RL   9  0    10K
   * Model statement for NPN transistors with model name QM
   .MODEL  QM NPN (IS=2E-16 BF=50 BR=1 RB=5 RC=1 RE=0 CJE=0.4PF
   +             VJE=0.8 ME=0.4 CJC=0.5PF VJC=0.8 CCS=1PF VA=100)
▲▲▲ *   Ac analysis for 10 Hz to 100 MHz with a decade increment and 10
    *   points per decade
```

```
    .AC   DEC   10   10   10MEGHZ
    .PLOT   AC   VM(9)   VP(9)
.END
```

The results of the frequency response are shown in Fig. 8-22. If the .PROBE command is included, there is no need for the .PLOT command.




**Figure 8-22** Frequency response for Example 8-7.

## Example 8-8

An astable multivibrator is shown in Fig. 8-23. The output is taken from nodes 1 and 2. Plot the transient responses of voltages at nodes 1 and 2 from 0 to 15 $\mu$s in steps of 0.1 $\mu$s. The initial voltages of nodes 1 and 3 are 0. The CPU time should




**Figure 8-23** Astable multivibrator.

be limited to 1.22E2 s. The model parameters of the BJTs are IS=2E−16, BF=50, BR=1, RB=5, RC=1, RE=0, TF=0.2NS, and TR=5NS.

**Solution** Due to the regenerative nature of the circuit, the solution may not converge, and the simulation will continue for a very long time. The CPU time is limited so that the circuit does not run for a long time. The run time should be less than the CPU time itself if the circuit converges. The listing of the circuit file follows.

**Example 8-8 Astable multivibrator**

```
▲ VCC  6  0  DC  5V
▲▲ RC1  6   1   1K
   RC2  6   2   1K
   R1   6   3   30K
   R2   6   4   30K
   C1   1   4   150PF
   C2   2   3   150PF
   *  Q1 and Q2 with model QM and substrate connected to ground by
   +default
   Q1   1   3   0   QM
   Q2   2   4   0   QM
   *  Model statement for NPN transistors
   .MODEL QM NPN (IS=2E-16 BF=50 BR=1 RB=5 RC=1 RE=0 TF=0.2NS TR=5NS)
   *   CPU time is limited.
   .OPTIONS  NOPAGE  NOECHO  CPTIME=1.2E2
   *   Node voltages are set to defined values to break the tie-in
   +condition.
   .NODESET V(1)=0  V(3)=0
▲▲▲ *   Transient analysis from 0 to 10 μs with 0.1-μs increment
    .TRAN/OP  0.1US  10US
    *   Plot the results of transient analysis: voltages at nodes 2 and +4.
    .PLOT  TRAN  V(1)  V(2)
    .OPTIONS  ABSTOL=1.0N  RELTOL=10M  VNTOL=1M  ITL5=40000
    .PROBE
.END
```

The transient responses are shown in Fig. 8-24. If the .PROBE command is included, there is no need for the .PLOT command.

## Example 8-9

A TTL inverter circuit is shown in Fig. 8-25(a). The output is taken from node 4. Plot the dc transfer characteristic V(4) versus $V_{in}$ if the input voltage is varied from 0 to 2 V with a step of 0.01 V. If the input is a pulse voltage with a period of 60 $\mu$s, as shown in Fig. 8-25(b), plot the transient response of voltage at node 4 from 0 to 80 ns in steps of 1 ns. The model parameters of the BJTs are BF=50, RB=70, RC=40, CCS=2PF, TF=0.1NS, TR=10NS, VJC=0.85, and VAF=50.

**Solution** The listing of the circuit file follows.

**Example 8-9 TTL inverter**

```
▲ *   Pulsed input voltage
  VIN  1  0  PULSE (0  5  1NS  1NS  1NS  38NS  60NS)
  VCC  6  0  DC  5V
```




**Figure 8-24** Transient responses for Example 8-8.




**Figure 8-25** A TTL inverter.

```
▲▲ *   BJTs with model QN and substrate connected to ground by default
   Q1   3  2  1  QN
   Q2   4  3  5  QN
   Q3   4  5  0  QN
   *   Model for NPN BJTs with model QN
   .MODEL QN NPN (BF=50 RB=70 RC=40 CCS=2PF TF=0.1NS TR=10NS VJC=0.85
   +VAF=50)
   R1   6  2  4K
   R2   5  0  1K
   R3   6  4  1K
▲▲▲ *  DC sweep for 0 to 2 with 0.01 V increment
      .DC  VIN  0  2  0.01
      *  Transient analysis for 0 to 80 ns with 1-ns increment
      .TRAN 0.5NS 80NS
      *  Plot the results of dc sweep: voltage at node 4 versus VIN.
      .PLOT  DC  V(4)
      *  Plot the results of transient analysis: voltage at nodes 4 and 1.
      .PLOT  TRAN  V(4)  V(1)
      .PROBE
.END
```

The results of the dc sweep and transient analyses are shown in Figs. 8-26 and 8-27, respectively. If the .PROBE command is included, there is no need for the .PLOT command.




**Figure 8-26** Dc transfer characteristic for Example 8-9.




**Figure 8-27** Transient response for Example 8-9.

## Example 8-10

A TTL inverter circuit is shown in Fig. 8-28(a). Plot the dc transfer characteristic between nodes 1 and 9 for values of $V_{in}$ in the range of 0 to 2 V in steps of 0.01 V. If the input is a pulsed waveform of period 80 $\mu$s, as shown in Fig. 8-28(b), plot the transient response from 0 to 80 ns with steps of 1 ns. The model parameters of the BJTs are BF=50, RB=70, RC=40, TF=0.1NS, TR=10NS, VJC=0.85, and VAF=50. The model parameters of the diodes are RS=40, TT=0.1NS.

**Solution** The listing of the circuit file follows.

**Example 8-10 TTL inverter**

```
▲ *  Pulse input voltage
  VIN  1  0  PULSE (0  3.5V  1NS 1NS  1NS  38NS  80NS)
  VCC  13  0  5V
▲▲ RS  1  2  50
   RB1 13  3  4K
   RC2 13  5  1.4K
   RE2  6  0  1K
   RC3  13  7  100
   RB5  13  10  4K
   *  BJTs with model QNP and substrate connected to ground by default
   Q1  4  3  2  QNP
   Q2  5  4  6  QNP
   Q3  7  5  8  QNP
   Q4  9  6  0  QNP
```




**Figure 8-28** TTL inverter.

```
Q5  11 10 9  QNP
*  Diodes with model DIODE
D1  8  9   DIODE
D2  11 12  DIODE
D3  12  0  DIODE
*  Model of NPN transistors with model QNP
.MODEL QNP NPN (BF=50 RB=70 RC=40 TF=0.1NS TR=10NS VJC=0.85 VAF=50)
*  Diodes with model DIODE
.MODEL DIODE D (RS=40 TT=0.1NS)
▲▲▲ *  Dc sweep from 0 to 2 V with 0.01 V increment
 .DC  VIN  0  2  0.01
 *  Transient analysis from 0 to 80 ns with 1-ns increment
 .TRAN  1NS  80NS
 *  Plot the results of dc sweep: voltage at node 9 against VIN.
 .PLOT  DC  V(9)
 *  Plot the results of transient analysis: voltage at node 9.
 .PLOT  TRAN  V(9)
 .PROBE
.END
```

The results of the dc sweep and transient analyses are shown in Figs. 8-29 and 8-30, respectively. If the .PROBE command is included, there is no need for the .PLOT command.

## Example 8-11

The circuit diagram of an OR/NOR gate is shown in Fig. 8-31(a). The inputs to nodes 1 and 4 are pulses of period 60 $\mu$s, as shown in Fig. 8-31(b). Plot the transient responses of voltages at nodes 12, 13, and 1 from 0 to 100 ns in steps of 1 ns. The model parameters of the BJTs are BF=50, RB=70, RC=40, TF=0.1NS, TR=10NS, VJC=0.85, and VAF=50. The parameters of the diodes are RS=40, and TT=0.1NS.

**Solution** The listing of the circuit file follows.

**Example 8-11 OR/NOR logic gate**

```
▲ *  Pulsed input voltages
  VA  1  0  PULSE (0  -5  1NS  1NS  1NS  38NS  60NS)
  VB  4  0  PULSE (0  -5  1NS  1NS  1NS  38NS  60NS)
  VEE  0  14  DC  5.2V
▲▲ *   BJTs with model QN and substrate connected to ground by default
    Q1   5  4  3   QN
    Q2   7  8  3   QN
    Q3   2  1  3   QN
    Q4   0  9  8   QN
    Q5   0  2  13  QN
    Q6   0  7  12  QN
```




**Figure 8-29** Dc transfer characteristic for Example 8-10.




**Figure 8-30** Transient response for Example 8-10.




**Figure 8-31** OR/NOR logic gate.

```
.MODEL QN NPN (BF=50 RB=70 RC=40 TF=0.1NS TR=10NS VJC=0.85 VAF=50)
*  Diodes with model DIODE
D1   9  10  DIODE
D2   10  11  DIODE
.MODEL DIODE  D (RS=40  TT=0.1NS)
R1   0  2  220
R2   0  7  245
R3   3  14  779
R4   4  14  15K
R5   1  14  15K
R6   8  14  6.1K
R7   0  9   907
R8   11  14  4.98K
RLO  12  14  10K
RLN  13  14  10K
▲▲▲ *  Transient analysis from 0 to 80 ns with 1-ns increment
   .TRAN  0.5NS  120NS
   * Plot the results of transient analysis: voltages at nodes 12 and +13.
   .PLOT TRAN V(12) V(13) V(1)
   .PROBE
.END
```

The results of the transient analysis are shown in Fig. 8-32. If the .PROBE command is included, there is no need for the .PLOT command.




**Figure 8-32** Transient response for Example 8-11.

## SUMMARY

The statements for BJTS are

```
Q(name)  NC  NB  NE  NS  QNAME  [(area) value]
.MODEL  QNAME  NPN  (P1=V1 P2=V2 P3=V3 ........PN=VN)
.MODEL  QNAME  PNP  (P1=V1 P2=V2 P3=V3 ........PN=VN)
```

## REFERENCES


1. H. K. Gummel and H. C. Poon, "An integral charge control model for bipolar transistors," *Bell System Technical Journal*, Vol. 49, January 1970, pp. 827–852.
2. Ian Getreu, *Modeling the bipolar transistor*—Part # 062-2841-00. Beaverton, Ore.: Tektronix, Inc., 1979.
3. J. J. Ebers and J. J. Moll, "Large-signal behavior of junction transistors," *Proc. IRE*, Vol. 42, December 1954, pp. 1161–1172.
4. L. W. Nagel, *SPICE2—A computer program to simulate semiconductor circuits*, Memorandum no. ERL-M520, May 1975, Electronics Research Laboratory, University of California, Berkeley.
5. A. S. Grove, *Physics and Technology of Semiconductor Devices*. New York: Wiley, 1967.
6. R. B. Schilling, "A bipolar transistor model for device and circuit design," *RCA Review*, Vol. 32, September 1971, pp. 339–371.
7. S. Natarajan, "An effective approach to obtain model parameters for BJTs and FETs from data books," *IEEE Transactions on Education*, Vol. 35, No. 2., 1992, pp. 164–169.
8. M. H. Rashid, *SPICE For Power Electronics and Electric Power*. Englewood Cliffs, N.J.: Prentice Hall, 1993, Chapter 11.


## PROBLEMS

**8-1.** For Example 8-2, calculate the coefficients of a Fourier series for the output voltage.

**8-2.** For Example 8-6, calculate the equivalent input and output noise.

**8-3.** For example 8-7, plot the output impedance and the current gain.

**8-4.** For Fig. 8-28, calculate the input and output noise for frequencies from 1 Hz to 10 kHz.

**8-5.** For Fig. 8-28, calculate and plot the frequency response of the output voltage from 10 Hz to 10 MHz in decade steps with 10 points per decade. Assume the peak input voltage is 5 V. The model parameters of the BJTs are BF=50, RB=70, RC=40, TF=0.1NS, TR=10NS, VJC=0.85, and VAF=50. The model parameters of the diodes are RS=40, and TT=0.1NS.

**8-6.** For the circuit in Fig. P8-6, calculate and plot (a) the magnitude and phase angle of voltage gain, (b) the magnitude of input impedance, and (c) the magnitude of output impedance. The frequency is varied from 1 Hz to 10 MHz in decade steps with 10 points per decade. The peak input voltage is 10 mV. The model parameters of the

BJT are IS=2E−16, BF=50, BR=1, RB=5, RC=1, RE=0, CJE=0.4PF, VJE=0.8, ME=0.4, CJC=0.5PF, VJC=0.8, CCS=1PF, and VA=100.




**Figure P8-6**

**8-7.** Repeat Problem 8-6 for the circuit in Fig. P8-7.




**Figure P8-7**

**8-8.** Repeat Problem 8-6 for the circuit in Fig. P8-8.




**Figure P8-8**

**8-9.** Repeat Problem 8-6 for the circuit in Fig. P8-9.




**Figure P8-9**

**8-10.** Repeat Problem 8-6 for the circuit in Fig. P8-10. Calculate the input and output noise.




**Figure P8-10**

**8-11.** For the circuit in Fig. P8-11, calculate and print the dc transfer function (the voltage gain, the input resistance, and the output resistance) between the output current and the input voltage $V_{EE}$. The model parameters of the BJTs are BF=100, BR=1, RB=5, RC=1, RE=0, VJE=0.8, and VA=100.




**Figure P8-11**

**8-12.** For the circuit in Fig. P8-12, calculate and print the voltage gain, the input resistance, and the output resistance. The input voltage is 5 V dc. The model parameters of the BJTs are BF=100, BR=1, RB=5, RC=1, RE=0, VJE=0.8, and VA=100.




**Figure P8-12**

**8-13.** Use PSpice to perform a Monte Carlo analysis for six runs and for the dc sweep of Problem 8-11. The model parameter is $R=1$ for resistors. The circuit and transistor parameters having uniform deviations are

$R_1 = 12.5\ \text{k}\Omega \pm 5\%$
$R_C = 10\ \text{k}\Omega \pm 15\%$
$B_F = 100 \pm 50$
$V_A = 100 \pm 20$

**(a)** The greatest difference from the nominal run is to be printed.
**(b)** The maximum value of the output voltage is to be printed.
**(c)** The minimum value of the output voltage is to be printed.
**(d)** The first occurrence of the output voltage crossing below 5 V is to be printed.

**8-14.** Use PSpice to perform the worst-case analysis for Problem 8-13.

**8-15.** Use PSpice to perform a Monte Carlo analysis for five runs and for the dc sweep of Problem 8-12. The transistor parameters having uniform deviations are

$$B_F = 100 \pm 50$$
$$V_A = 100 \pm 20$$

**(a)** The greatest difference from the nominal run is to be printed.
**(b)** The maximum value of the output voltage is to be printed.
**(c)** The minimum value of the output voltage is to be printed.

**8-16.** Use PSpice to perform the worst-case analysis for Problem 8-15.